

ブーディン構造から伸長歪を求めるプログラム

増田俊明¹・渋谷朝紀²・森川知行³・大浦坂勝利⁴・Tran Ngoc NAM (トラン, N. ナム)¹

Calculation of extension strain from boudin structure: a computer program

Toshiaki MASUDA¹, Tomoki SHIBUTANI², Tomoyuki MORIKAWA³, Katsutoshi OHURASAKA⁴,
and Tran Ngoc NAM¹

Abstract This paper presents a computer program of a strain reversal method for measuring extension strain by boudin structures. The fundamental data for this method consists of the length of each boudin and interboudin gap distance. The principle of the method and the usage of the program are introduced with an example.

Key words: boudin structure, extension strain, computer program

要旨 ブーディン構造から伸長歪を求めるプログラムを開発した。測定するのはすべてのブーディンの長さ、各々のブーディン間の距離である。歪は strain reversal method に基づいており、歪を求める原理、プログラムの内容を示し、BASIC で書かれたプログラムを掲載した。

はじめに

塑性変形の激しさを、露頭あるいは薄片を観察して直感的に判断することは、経験を積んだ構造地質学者にとっては、ある程度は可能であると思われるが、誰にでもできる訳ではなく、ましてや、必ずしも客観的ではない。それよりも、個々の変形構造に対して変形量を定量化して数値で表現する方が、より客観的であり、確実に変形の激しさを知ることができる。現在までに、変形量を定量的に求める方法はいくつか考案されており(例えば、Ramsay & Huber 1983)、それらを利用して研究を進めるのが簡便である。また、最近のコンピュータの発達と普及に伴い、必要なデータをインプットすれば、然るべき変形量の値が出力されるプログラムも市場に出回り始めている(例えば、Macintosh の Rf/f method)。プログラムは信用できるものが一つあれば、これを利用した方が便利であり、全ての研究者が独自のプログラムをそれぞれ開発するのは時間の無駄

である。

増田と渋谷は、紅れん石のマイクロブーディン構造の応力-歪解析(Masuda & Kuriyama 1988, Masuda *et al.* 1989, 1990)の途上で、標題に示したプログラムを開発した。そのプログラムは NEC-98 シリーズ用に N88-BASIC で書かれている。大浦坂は、そのプログラムを Macintosh 用 Quick BASIC に変換した。森川は、さらにそのプログラムを PASCAL で書き換えた。Nam はこのプログラムを用いて、解析を行った。小文ではこれらのプログラムを公開し、必要な方がおられた場合には、ブーディン構造を利用して簡単に伸長歪が計算できる便宜を提供しようとするものである。

伸長歪を求める原理

歪を求める原理は、Ferguson (1981) の strain reversal method に基づいている。ただし、若干の改良

¹静岡大学理学部地球科学教室 422 静岡市大谷836

Institute of Geosciences, School of Science, Shizuoka University, 836 Oya, Shizuoka, 422 Japan.

²動力炉・核燃料開発事業団 319-11 茨城県那珂郡東海村

Power Reactor & Nuclear Fuel Development Corporation, Tokai-mura, Naka-gun, Ibaraki-ken, 319-11 Japan.

³東京大学教養学部宇宙地球科学教室 153 東京都目黒区駒場3-8-1.

Department of Earth and Planetary Sciences, University of Tokyo, 3-8-1 Komaba, Meguro-ku, Tokyo, 153 Japan.

⁴東日 410 沼津市大岡2240-16.

Tonichi Co. Ltd., 2240-16 Ooka, Numazu, 410 Japan.




図1 1つの鉱物粒子が破断し、切り離されときの歪の求め方 (Ferguson 1981 の Fig. 1 より抜粋). a: 変形前の状態, b: 変形後の状態. 歪 (ϵ) は $\epsilon = \ln \{(l + \delta l) / l\}$ で求める. l は両ブーデインの中点間の距離である.

点が含まれている. 変形は純粋剪断変形 (pure shear) であるとする. また, ブーデインを構成する物質は, マトリクスに対してじゅうぶんに剛体的に振るまい, 変形の際に, 形を変えずに, 単に切り離されるだけであるとする. 以下はブーデインを構成する物質が柱状の鉱物粒子であることを想定して記述してあるが, それが板状の鉱物であったり, 岩石や地層であっても原理は同じである.

図1のa及びbには, 柱状の鉱物の伸びの方向 (長軸方向) に伸張が起こり, 柱状鉱物が破断し, 離れる様子が示してある. この場合の離れた方向での伸長歪を検討する. 歪はブーデイン間の距離 (図1の区間B-C) の変化に基づいて求めることができる. ただし, BとCはそれぞれのブーデインの中央に位置している. すなわち, 変形前は l であった長さが, 変形後には $l + \delta l$ になっているので, 歪 (natural finite strain) (ϵ) は

$$\epsilon = \ln \{(l + \delta l) / l\} \quad (1)$$

である. ここでは, もともとの長さが微小歪 ($d\epsilon$) を受けた後の長さの変化 ($d\ell$) を考える. すなわち

$$d\epsilon = d\ell / l$$

これを積分すれば (1) 式を導くことになり, ϵ は求めることができるが, ここでは $d\epsilon$ の代わりに微小ではあるが有限の歪 $\Delta\epsilon$ を導入し, 歪を近似的に求めることを考える. すなわち

$$\Delta\epsilon = \Delta\ell / l$$

と表現できる. ここで, l は $\Delta\epsilon$ だけ変形する前の長さ, $\Delta\ell$ は $\Delta\epsilon$ だけ変形する間に変化した長さである. $\Delta\epsilon$ だけ変形した後の長さは $l + \Delta\ell$ となる. この式を書き換えて, 変形後の長さを変形前の長さを表すようにすると

$$l = (l + \Delta\ell) / (1 + \Delta\epsilon)$$

となる. この式は, $l + \Delta\ell$ の長さのものが, $\Delta\epsilon$ だけ歪む前には, 長さが l であったことを意味する.

今, 図1bのようなブーデイン構造について, $\Delta\epsilon$ だけ歪を仮想的に戻してやることを考える. $l + \delta l$ が歪を戻す前の長さで, $\Delta\epsilon$ だけ歪を戻した時の長さ l_1 は

$$l_1 = (l + \delta l) / (1 + \Delta\epsilon)$$

となる. さらに, また $\Delta\epsilon$ だけ歪を戻した時の長さ l_2 は

$$l_2 = l_1 / (1 + \Delta\epsilon)$$

である. 以下, 同様に歪を $\Delta\epsilon$ だけ戻していくと, 両ブーデイン間の距離は徐々に短くなり, n 回目で両者が接触するまでになる. この時の状態が図1aに示されている. すなわち, ブーデイン化が起こる前の状態に戻ったことになる. 図1bから図1aまで戻すのに必要な歪は, 上記の計算の回数 n と $\Delta\epsilon$ の大きさから

$$\epsilon = n\Delta\epsilon$$

と求めることが出来る. その ϵ の値こそが, 図1aの状態から図1bの状態にまで変形したときの全歪量である. ただし, この方法では, 初めて粒子が割れてからの歪が計算される. その粒子が割れる前に岩石が被っていた歪については, この方法では知ることは出来ない. 現在歪0の状態として, そこから過去へ遡って歪を測っている, ということなので, strain reversal method と称されているのである.

$\Delta\epsilon$ の値については, あまり大きすぎると, 実際の歪の値の近似にならなくなり, また, 小さすぎると多数の計算回数が必要になり, 計算誤差や計算時間のことが気になるようになる. 我々の場合には, $\Delta\epsilon = 0.0001$ の値を使っている. 例えば図1にあるブーデイン構造の場合に, 実際に積分をして求めた値と, この方法で数値的に求めた値とでは, 0.01% 以下の差しかなく, この程度の誤差は, ほとんど問題にならない. 計算時間にしても, 特に長いと感じるほどの時間ではない.

1つの粒子が多数のブーデインに別れた場合にも, 各々の割れ目の両側のブーデインを一つの単位として捉えて, 基本的に同様の計算を行えば, 各々の割れ目ができた後の歪を求めることができる. ただしこの場合には, 複数の割れ目は厳密にいつ同時ににはできないという仮定で歪を見積もっていることに注意を要する. すなわち, このことは, 徐々に歪を戻していくと, 順次割れ目が閉じていき, 最後の割れ目が閉じた時に, 元の形が復元できることを意味する.

プログラムの利用法

補遺1にプログラムを示し, 補遺2と図2にそのプログラムを用いた解析例を示す. 現在観察されるブーデイン構造 (図2a) から, ブーデイン形成前の状態 (図2e) を復元する. 測定するのはブーデインの長さ




図2 ブーディン構造から伸長歪を求める解析例。現実に観察されるブーディン構造 (a) から、破断して切り離される前の状態 (e) を復元する。測定するのは a に示されているブーディンの長さ $B_{1a} - B_{5a}$ と離された距離 $G_{1a} - G_{4a}$ である。ブーディンの中間点の距離 $L_{1a} - L_{4a}$ は測定する必要はない。ブーディン内の破線は、そこで破断が起こったことを示している。a の状態から b の状態までに受けた歪は 29.19 %、b の状態から c の状態までに受けた歪は 21.89 %、c の状態から d の状態までに受けた歪は 7.47 %、d の状態から e の状態までに受けた歪は 8.42 % であり、プリンター出力の k として表示される (補遺 2)。破断した場所についてはプリンター出力の t を参照されたい (補遺 2)。t の値は左から数えて何番目の割れ目が閉じたのかを示している。a の状態からの総歪量についてはプリンター出力 (補遺 2) の TOTAL STRAIN を参照のこと。

($B_{1a} \sim B_{5a}$) と、切り離された距離 ($G_{1a} \sim G_{4a}$) である。図 2 の場合には、データ文 (補遺 1 のプログラムの 1 0 1 0 行) に、まず、粒子の番号を入力する。この場合には 1 である。これは、多数の粒子を同時に処理するのに必要な操作である。次に入力するのは割れ目の数である。これは、プログラムを走らせる上で必要なデータである。図 2 の場合には 4 である。4 と粒子の番号の 1 の間にはカンマが必要である。その後ブーディンの長さとしてブーディン間の距離を順次入力していく (入力例; $B_{1a}, G_{1a}, B_{2a}, G_{2a}, \dots, B_{4a}, G_{5a}$ の順)。この例の場合には図の左側から順に入力してある。そして最後に、ブーディンの幅を入力する。これは、歪を求めるのには直接関係はないが、破断が起こった時の粒子 (層) の縦横比を知るのに必要である。各々の数値の間にはカンマを入れる。これで入力完了する。もしも、別の粒子 (層) のデータを同時に処理したい場合には、新たにデータ文を設けて (プログラムの 1 0 2 0 行以下)、先ほどと同じ順でその粒子のデータを入力すればよい。このとき、前の粒子とは別の粒子番号を付けておく。

実際に計算を実行させると、モニター画面の左側に数字が順次出力されていく。この数字は計算回数を示している。コンピュータが高速の場合には、次々と数字が移り変わって行く様子が見える。やがて、あるブーディン間の距離がゼロ (計算上は微小なマイナスの値) になると、モニター画面にはその時まで計算した歪の大きさ (横軸) と、距離がゼロになった二つのブーディンが合体した状態の縦横比 (縦軸) との関係が、小さな円で表示される。この際、プリンターには、 $L_{1b}, L_{2b}, \dots, L_{4b}$ 及びすべてのブーディン間の距離のデータが出力される (補遺 2)。また、左側から数えて何番目の割れ目が閉じたのかも出力される (プリンター出力では t に相当する: 補遺 2)。図 2 の例では、a から b までの歪は 21.19 %、閉じた割れ目は左から 4 番目である。この段階で、閉じた割れ目は無いものとして自動的

に再登録され、改めて全てのブーディンの長さとして、ブーディン間の距離が計算され (図 2 の B_{1b}, \dots, B_{4b} と G_{1b}, \dots, G_{3b} を意味する)、出力される。この時、ブーディンの数が一つ減っている。この段階で、再び歪を元に戻す計算が開始され、モニターには数字が次々と打ち出されていく。このプロセスを繰り返して実行すると、やがてすべてのブーディン間の距離がゼロになる。この段階で一つの粒子の解析は終了する。このようにして得られた結果に基づいて、歪を戻していくときにどのようにして割れ目が閉じていったのかを図示したものが図 2 である。図を描くプログラムはまだ未完成で、ここは手作業である。この図により、1 つの鉱物粒子 (層) が 5 つのブーディンに分かれていった経緯を知ることができる。別の粒子のデータが入力されている場合には、その粒子について、同様の計算が引き続き行われることになり、すべての粒子について結果が出力された段階でプログラムは終了する。

付 記

BASIC で書かれたプログラムを補遺に掲載した。なお、PASCAL で書かれたプログラムは割愛した。このプログラムは商業用ではないので、計算が正しく行われればとりあえず問題無しとして、必ずしも合理的かつ経済的に組まれているわけではない。特に出力に関してはほとんど工夫をこらしていない。したがって、あらゆるニーズに応える形式にはなっていないので、利用者の独自のニーズで工夫を加えてプログラムを改良して頂きたい。なお、Macintosh 用のプログラムに興味をお持ちの方は、筆者 (森川または増田) まで連絡して下さい。

謝 辞

故鯨島輝彦教授は、マイクロブーディン構造の研究の良き理解者でした。我々の研究を支えていただきましたことに深く感謝いたします。なお、この原稿は東京大学の吉田鎮男助教授と静岡大学の長濱裕幸博士に査読して頂きました。

引用文献

- FERGUSON C. C. (1981), A strain reversal method for estimating extension from fragmented rigid inclusions. *Tectonophysics*, **79**, T43-T52.
- MASUDA T. & KURIYAMA M. (1988), Successive "mid-point" fracturing during microboudinage: an estimate of the stress-strain relation during a natural deformation. *Tectonophysics*, **147**, 171-177.
- MASUDA T., SHIBUTANI T., IGARASHI T. & KURIYAMA M. (1989), Microboudin structure of piedmontite in quartz schists: a proposal for a new indicator of relative palaeodifferential stress. *Tectonophysics*, **163**, 169-180.
- MASUDA T., SHIBUTANI T., KURIYAMA M. & IGARASHI, T. (1990), Development of microboudinage: an estimate of changing differential stress with increasing strain. *Tectonophysics*,

178, 379-387.

RAMSAY J. G. & HUBER M. I. (1983), *Modern Structural Geology*, Academic Press, London, 307p.

補遺1 伸長歪を求めるプログラム。プログラム言語はBASICを用いている。

```

10 ' STRAIN REVERSAL METHOD BY C. C. FERGUSON
20 ' STRAIN REVERSAL METHOD MODIFIED BY T. MASUDA
30 ' PIED(I) : LENGTH OF SEPARATED PIEDMONTITE BOUDIN
40 ' SEP(I) : LENGTH OF SEPARATION BETWEEN BOUDINS
50 ' N : NUMBER OF SEPARATION
60 CLS 3
70 SCREEN 3
80 CONSOLE 0,25,0,1
90 LINE (0,380)-(630,380) : LINE (630,380)-(630,0)
100 FOR I=1 TO 30
110 LINE (630-20*I,370)-(630-20*I,380)
120 NEXT I
130 LINE (620,180)-(630,180)
140 PPP=203 : ' PPP is the total number of boudinaged grains
150 FOR SSS=1 TO PPP STEP 1
160 KKK=1 : TOTAL=0
170 READ NUM
180 'lprint "No = ";num
190 READ N
200 DIM L(N),L0(N),PIED(N+1),SEP(N)
210 FOR I=1 TO N
220 READ PIED(I)
230 READ SEP(I)
240 NEXT I
250 READ PIED(N+1)
260 READ W
270 ' if n>1 goto 570
275 LPRINT "No = ";NUM
300 LPRINT:;LPRINT:LPRINT
310 '
320 LPRINT "KKK = ";KKK
330 FOR I=1 TO N
340 L0(I)=(PIED(I)+PIED(I+1))/2
350 NEXT I
360 FOR I=1 TO N
370 L(I)=L0(I)+SEP(I)
380 NEXT I
390 FOR I=1 TO N
400 LPRINT "L0(" ; I ; ")=" ; L0(I) , "L(" ; I ; ")=" ; L(I)
410 NEXT I
420 LPRINT
430 '
440 '
450 FOR K=1 TO 1E+06
460 PRINT "K = ";K

```

```

470 '
480   FOR I=1 TO N
490   L(I)=L(I)/1.0001      :' 0.01% strain step
500   SEP(I)=L(I)-L0(I)
510 '   print "L("i;")=";l(i), "sep(";i;")=";sep(i)
520 NEXT I
530   FOR I=J TO N
540   IF SEP(I)<0 THEN 610
550   NEXT I
560 NEXT K
570 LPRINT : I.PRINT
580 ERASE L,L0,PIED,SEP
590 NEXT SSS
600 END
610 '
620 T=I : R=L0(I)/W*2
630 TOTAL=TOTAL+K
640 LPRINT "k = ";K,"t = ";T,"aspect ratio = ";R
650 LPRINT
660 LPRINT "                                TOTAL STRAIN =" ;TOTAL.
670 CIRCLE (630-.1*TOTAL.,380-20*R),4,4
680 LPRINT
690 FOR I=1 TO N
700 LPRINT "L(";I;")=";L(I),"SEP(";I;")=";SEP(I)
710 NEXT I
720 LPRINT : I.PRINT : I.PRINT
730 FOR I=1 TO N-1
740 IF I<T THEN PIED(I)=PIED(I):SEP(I)=L(I)-L0(I):GOTO 770
750 IF I=T THEN PIED(I)=PIED(I)+PIED(I+1):SEP(I)=SEP(I+1):GOTO 770
760 IF I>T THEN PIED(I)=PIED(I+1):SEP(I)=SEP(I+1)
770 NEXT I
780 IF N>T THEN PIED(N)=PIED(N+1)
790 IF N=T THEN PIED(N)=PIED(N)+PIED(N+1)
800 KKK=KKK+1
810 N=N-1
820 IF N=0 THEN 570
830 GOTO 310
1000 'DATA 2,4,14,10,16,20,17.5,15,10,5,20,5
1010 DATA 1,4,13.5,10,16.5,20,16.5,15,10,5,19.5,5

```

補遺2 解析例 (プリンターの出力例)

No = 1

KKK = 1

L(1) = 15 I(1) = 25

L(2) = 16.5 L(2) = 36.5

L(3) = 13.25 I(3) = 28.25

L(4) = 14.75 L(4) = 19.75

k = 2919 t = 4 aspect ratio = 5.9

TOTAL STRAIN = 2919

L(1) = 18.6704

SEP(1) = 3.67044

L(2) = 27.2589

SEP(2) = 10.7589

L(3) = 21.0976

SEP(3) = 7.8476

L(4) = 14.7497

SEP(4) = -3.43323E-04

KKK = 2

L(1) = 15 I(1) = 18.6704

L(2) = 16.5 I(2) = 27.2589

L(3) = 23 I(3) = 30.8476

k = 2189 t = 1 aspect ratio = 6

TOTAL STRAIN = 5108

L(1) = 14.9995

SEP(1) = -5.09262E-04

L(2) = 21.8993

SEP(2) = 5.39928

L(3) = 24.7824

SEP(3) = 1.7824

KKK = 3

L(1) = 23.25 I(1) = 28.6493

L(2) = 23 L(2) = 24.7824

k = 747 t = 2 aspect ratio = 9.2

TOTAL STRAIN = 5855

L(1) = 26.5869

SEP(1) = 3.33694

L(2) = 22.9984

SEP(2) = -.0015831

KKK = 4

L(1) = 38 I(1) = 41.3369

k = 842 t = 1 aspect ratio = 15.2

TOTAL STRAIN = 6697

L(1) = 37.9985

SEP(1) = -1.49536E-03